

# Modelling Bitcoin in Agda\*

Anton Setzer<sup>1</sup>

- 1 Dept. of Computer Science, Swansea University, Singleton Park, Swansea SA2 8PP, UK  
a.g.setzer@swansea.ac.uk

---

## Abstract

We present two models of the block chain of Bitcoin in the interactive theorem prover Agda. The first one is based on a simple model of bank accounts, while having transactions with multiple inputs and outputs. The second model models transactions, which refer directly to unspent transaction outputs, rather than user accounts. The resulting blockchain gives rise to a transaction tree. That model is formalised using an extended form of induction-recursion, one of the unique features of Agda. The set of transaction trees and transactions is defined inductively, while simultaneously recursively defining the list of unspent transaction outputs. Both structures model standard transactions, coinbase transactions, transaction fees, the exact message to be signed by those spending money in a transaction, block rewards, blocks, and the blockchain, and the second structure models as well maturation time for coinbase transactions and Merkle trees. Hashing and cryptographic operations and their correctness are dealt with abstractly by postulating corresponding operations. An indication is given how the correctness of this model could be specified and proven in Agda.

**1998 ACM Subject Classification** F.3.1: Specifying and Verifying and Reasoning about Programs; I.6.5: Model Development; D.2.4: Software/Program Verification; D.1.1: Applicative (Functional) Programming; K.4.4: Electronic Commerce

**Keywords and phrases** Blockchain; cryptocurrency; Bitcoin; Agda; verification; smart contract

## 1 Introduction

Since its introduction in November 2008, the market capitalisation of bitcoins has risen to over 161 Billion US-\$ (as of 27 November 17) [12]. In 2017 the price for one bitcoin has risen from 1001 US\$ on January 1 to 9665 US\$ on 27 Nov 2017 [12]. Other cryptocurrencies have been introduced as well, the main contender being currently Ethereum with a market capitalisation of almost 46 Billion US-\$. Cryptocurrencies have become a major financial instrument and might even become major currencies in the near future.

Cryptocurrencies have as well been used for the use in smart contracts [32, 38]. The simplest form of a smart contract is where the buyer reserves money for the seller on the blockchain. In order to unlock the money, the seller needs to receive a second signature from the buyer, which is given once the goods are received. If the goods don't arrive on time, the money is returned to the buyer. Smart contracts are now used in large funds, where all contractual relationships are entirely governed by algorithms with no legal framework being used.

---

\* Work supported by EPSRC grant EP/G033374/1 *Theory and Applications of Induction Recursion*; CORCON FP7 Marie Curie International Research Project, PIRSES-GA-2013-612638; COMPUTAL FP7 Marie Curie International Research Project, PIRSES-GA-2011-294962; CA COST Action CA15123 (EUTYPES)

Since cryptocurrencies are entirely governed by algorithms, one needs to have a very high guarantee that these algorithms are correct – in case of failure there is no legal framework available to remedy the problem. The only way to fix problems is to find a consensus amongst users and amongst miners<sup>1</sup> to change the protocol and create a soft (i.e. backward compatible) or hard fork. A possible mistake in the algorithms is one of the most important risks in cryptocurrencies. Some mistakes and weaknesses have already been found, of which some have been fixed. In the original Bitcoin protocol, uniqueness of transaction IDs was not guaranteed, which would allow replay attacks. This could be fixed by including the block number in coinbase transaction by a soft protocol change (see the discussion in Subsect. 7.2). The size of the blocks of Bitcoin is no longer enough to cover all transactions, a problem which is known, but which couldn't be solved at time of writing this article: Bitcoin cash [34] would have been a solution, but became an alternative currency instead of being a corrected form of Bitcoin; the SegWit soft fork [36] was successful, but not enough to fix the problem; the second stage of SegWit, SegWit2x [37] was cancelled on 8 Nov 2017 [7].

Therefore, it is important to fully verify cryptocurrency protocols. A very high level of verification can be achieved by creating formal models for cryptocurrency protocols, and proving their correctness. We haven't found a fully worked out complete formal model and correctness proof of cryptocurrency protocols, and the goal of this article is to take first steps in order to fill this gap.

The rise of smart contracts has led to new problems concerning the security of financial investments. The up to now biggest incident was the failure of the DAO [35], a form of investor-directed venture capital fund based on smart contracts in the cryptocurrency Ethereum. Malicious users exploited a vulnerability in the DAO, at a time when the market value of the DAO had reached 150 Million US-\$. The loss of the investor's money was only avoided by making a hard fork, which deleted most transactions investing in the DAO. This hard fork violated the principle that control over the money owned by users in cryptocurrencies should only be governed by algorithms, without any human intervention. Therefore, there is need for smart contracts, which are given together with a correctness argument, ideally a formal proof, which serves as a certificate. Such proofs need to rely on a model of the underlying cryptocurrency.

In this article we will develop, as a first step towards verifying the protocols of cryptocurrencies and developing verified smart contracts, a model of Bitcoin in the interactive theorem prover and dependently typed programming language Agda [4, 3, 31]. We will formalise how notions of transaction trees, transactions, blocks, unspent transactions, and how messages are signed. We will as well include aspects such as coinbase transactions, giving the miner their rewards, and maturation time for coinbase transactions. We will not include all aspects yet, especially input and output scripts and the proof of work will not be part of the model at this stage, although we assume that including them would be relatively straightforward. We will indicate how correctness of Bitcoin can be specified, although we will leave it to a future paper to actually carry out the precise specification and proof of correctness. This seems at the moment more a matter of using enough man power to carry out those proofs in an interactive theorem prover, rather than solving fundamental problems.

What we will see is that the blockchain is from a theoretical point of view a more complicated data structure than expected. Our definition will simultaneously define the set of transaction trees and transactions together with the unspent transactions, which will be an extended form of an inductive-recursive definition [14, 16, 15]. This could be one reason

---

<sup>1</sup> In case of soft forks there are user activated (UASF) vs miner activated (MASF) soft forks.

(apart from the fact that Bitcoin is rather new – the announcement and mining of the first block happened on 1 November 2008 [26, 25]), why descriptions of Bitcoin are often very difficult to understand. Many details are well described, but descriptions of transactions which take into account that due to the reference to unspent transaction outputs one obtains a tree like structure, often lack clarity. We hope that the current paper will help to understand this data structure better. We hope as well that it will help to popularise the concept of induction-recursion, which we believe is a concept, which is not restricted to dependent type theory and Agda, but can be of great use in a more general mathematical and computer science setting.

**Content of this Article.** We will first give in the next Sect. 2 a brief introduction into Agda. Then we show in Sect. 3, mainly following [33], how to develop, starting from a simple model of a bank, a ledger based model of Bitcoin. A model in Agda based on this is presented in Sect. 4. This definition is much simpler than the final model, since we can define ledgers and transaction separately, without the need of induction-recursion. In Sect. 5 we discuss the security problems of this model, and introduce in theoretical form the notions of transaction trees, transactions, and unspent transaction outputs. In Sect. 6 we develop a model in Agda which makes use of transaction trees, trees, and unspent transaction outputs. This model will be make use of inductive-recursion. In Sect. 7 we discuss how one could specify and prove the correctness of this model, and as well limitations of our approach (not using input and output scripts, and the need to add the miners’ proof of work to the model). We finish by a short conclusion including future work (Sect. 8).

**Related Work.** We haven’t found any research on developing the transaction tree as a formal structure. A lot of research has been carried out on modelling the Bitcoin and related protocols by modelling Bitcoin as interactions between agents. Beukema [8] introduces a model of Bitcoin based on transactions referring to previous ones. It is formalised in the language mCRL2 as a transition system between agents. An analysis is carried out using model checking regarding the behaviour of the network, and how the consensus protocol deals with corrupted messages and double spending attacks. Andrychowicz et. al. [5] model Bitcoin as timed automata and verify using UPPAAL correctness properties. The automata formalise how agents interact regarding the original Bitcoin protocol. Chaudhary, et. al. [11] model the interactions in Bitcoin as a state transition system between agents. They use the model checker UPPAAL, and analyse the probability that a malicious transaction is included in the longest chain, which could give rise to double spending. Bastiaan [6] provides a stochastic analysis of the Bitcoin mining protocol modelled using continuous-time Markov chains. The focus of that paper is on analysing mining and how to prevent the formation of large pools of miners.

A lot of research has been carried out on verifying smart contracts. Kosba et. al. [20, 21] present a programming language Hawk for writing smart contracts. They verify privacy properties using formal methods. Their underlying model is a ledger model. Luu et. al. [22] analyse vulnerabilities in smart contracts. Their focus is directly on the language for smart contracts. Bhargavan et. al. [9] translate smart contracts in Ethereum into  $F^*$  and analyse them.

**Acknowledgement.** We would like to thank Andrew Lawrence for inspiring us to start the project of formalising Bitcoin. His idea led to a series of third year projects and MSc projects under the author’s supervision.. The first one, in which the main basis was formed, was

carried out by Damon Jones [19], which was followed<sup>2</sup> by third year and MSc projects by M. Zahid, Ifetayo Agunbiade, Ehsan Alebrahim-Dehkordi, Kieran Cullinan, Dylan MacLennan, Robert Locklan, and some projects, which are in their initial phase. These models were based on the brown bag talk by Warner [33]. The first Agda model in this article (Sect. 4) will be based on the models developed in those projects, with some enhancements such as adding the block reward, and the precise signatures for transactions. We want to thank those students for invaluable insights they added to this project.

**Agda Code.** Every line of Agda code in display format provided in this paper has been type-checked by Agda and rendered by the Agda  $\text{\LaTeX}$ -backend, and is therefore type safe. However, we mostly omit administrative parts of the code such as modules and name space handling; thus, the code as printed in this article will not be accepted by Agda as-is. The complete type checked code can be found in [29].

**Disclaimer.** This is a theoretical model, developed for scientific purposes only, which models some aspects of blockchain technology and Bitcoin. It is not intended as a tool to make business decisions. Business decisions should be based on other sources. The legal disclaimer from bitcoin.org (<https://bitcoin.org/en/legal>) will apply to this article, with “the website” replaced by “the current article (Anton Setzer: Modelling Bitcoin in Agda)”.

## 2 Introduction into Agda

Agda [4, 3, 31] is a theorem prover based on intensional Martin-Löf type theory [23]. It is closely related to the theorem prover Coq [13]. Furthermore, Agda is a total language, which is guaranteed by its termination and coverage checker – without it Agda would be inconsistent. The current version of Agda is Agda 2, which was initially designed and implemented by Ulf Norell in his PhD thesis [27], and has since been developed further by the Agda community.

In Agda, there are infinitely many levels of types: the lowest one is called for historic reasons `Set` – `Set` stands for what is in other languages called “type”. The main type constructions in Agda are dependent function types, inductive types, coinductive types (which don’t occur in this article), and record types.

Inductive data types are dependent versions of algebraic data types as they occur in functional programming. They are given as sets  $A$  together with constructors which are strictly positive in  $A$ . For instance, the even and odd numbers are given by the simultaneous — as denoted by the keyword `mutual` — inductive data types:

```
mutual
data Even : ℕ → Set where
  0p      : Even 0
  succp   : {n : ℕ} → Odd n → Even (suc n)

data Odd : ℕ → Set where
  succp   : {n : ℕ} → Even n → Odd (suc n)
```

The expression  $(n : \mathbb{N}) \rightarrow A$  denotes a dependent function type, which is similar to a function type, but  $A$  can depend on the argument  $n$ . The expression  $\{n : \mathbb{N}\} \rightarrow A$  is an implicit version of the previous construct. Implicit arguments can be omitted, provided they can be inferred by the type checker. We can make an implicit argument explicit by writing,

---

<sup>2</sup> I hope this list is complete.

e.g., `sucp {0} Op`. We can as well write `sucp {n = 0} Op`, which is useful when having several implicit arguments, of which only some are given explicitly. If there are several explicit or implicit dependent arguments in a type, one can omit “ $\rightarrow$ ”, as illustrated in the following example:  $(a : A)(b : B) \rightarrow C$  instead of  $(a : A) \rightarrow (b : B) \rightarrow C$ . The elements of `(Even n)` and `(Odd n)` are those that result from applying the respective constructors. Therefore, we can define functions by case distinction on these constructors using pattern matching, e.g.

```
mutual
  _+e_ : ∀ {n m} → Even n → Even m → Even (n + m)
  Op    +e p =      p
  suc p +e q =      suc (p +o q)

  _+o_ : ∀ {n m} → Odd n → Even m → Odd (n + m)
  suc p +o q =      suc (p +e q)
```

Here,  $\forall a \rightarrow B$  is an abbreviation for  $(a : A) \rightarrow B$ , where  $A$  can be inferred by Agda.  $\forall \{a\} \rightarrow B$  is the same but for an implicit argument, while  $\forall \{n m\} \rightarrow B$  abbreviates  $\forall \{n\} \rightarrow \forall \{m\} \rightarrow B$ . Agda supports infix operators, where “ $\_$ ” denotes the position of the arguments. For instance, `(Op +e p)` stands for `(_+e_ Op p)`. The combination of infix symbols together with the availability of Unicode symbols makes it possible to define Agda code which is very close to standard mathematical notation.

Nested patterns are allowed in pattern matching. The coverage checker verifies completeness and the termination checker ensures that the recursive calls follow a schema of extended primitive recursion. In Agda, inductive types have been substantially extended. Agda supports a highly generalised version of indexed inductive-recursive and inductive-inductive definitions. This allows to define simultaneously several sets inductively while defining simultaneously recursively functions.

An important indexed data type is propositional equality  $x \equiv y$  (for  $x, y : A$ ) which has as constructor a proof of reflexivity. It expresses that propositional equality is the least reflexive relation (modulo the built-in definitional equality of Agda):

```
data _≡_ {a} {A : Set a} (x : A) : A → Set a where
  refl : x ≡ x
```

An example of a record type is as follows:

```
record Student : Set where
  constructor student
  field name : String
        studnr : ℕ
open Student public
```

We will in the following omit in this paper the keyword `field`, and as well the part starting with `open` (which is needed in order to access the fields without prefixing them with the name of the record type). The part starting with `constructor` is optional, using it we can introduce an element of `Student` by writing `exampleStudent = student "John" 123456`. We can access the fields of a record by writing for instance `exampleStudent.name`. We can use as well the fields as functions, for instance `name : Student → String`.

Agda allows as well to assume the existence of an element of a type without defining it, for instance

`postulate hypotheticalStudent : Student`


This is consistent, as long as one could in principle define an element of that type. All occurrences of `postulate` in this article are of this form. In general `postulate` allows to make inconsistent definitions, and needs to be handled with care. An inconsistent example postulates the existence of an element of the empty set `Fin 0`:

`postulate nonExistentElement : Fin 0`

### 3 A Basic Ledger Based Model of Bitcoin

#### 3.1 From a Model of a Bank to a Bitcoin Ledger

We will in the following repeat briefly the steps taking in the brown bag talk by Warner [33] on how to derive a Bitcoin transaction tree starting from a simple model of a bank. From this model we will derive in Sect. 4 a model of the ledger in Agda. In Sect. 5 we will then adapt this model to take into account that inputs of transactions are unspent transaction outputs rather than part of the amount contained in the ledger. However, we will already here refine this model by adding coinbase transactions, a block reward function, and defining the signatures for a transaction.



**Figure 1** Source: [33] (Licensed under creative common license <https://creativecommons.org/licenses/by-sa/3.0/>)

One starts (Fig. 1 (a)) with a simple model of a bank which consists of a ledger and transaction. The ledger is a table, which determines for each time (taken as natural numbers) and account the amount in this account at that time. Whereas in banks usually transactions are from one account to another account. this is already at this time generalised to transactions having multiple inputs and outputs. There can as well be transactions without inputs, which corresponds to money created by the central bank. In case of Bitcoin, it will later be replaced by the block reward as given by a coinbase transaction. Transaction can as well loose money, which means their sum of outputs is less than the sum of inputs, which will in the Bitcoin protocol become transaction fees given to the next miner.

The next step is to hide the account owners behind public/private key pairs. The recipients of transactions are public keys, whereas the transaction needs to be signed by the private keys of the input accounts. Therefore, transactions are now controlled by the clients, not by the bank, but bank is still needed as a central server to control and synchronise the history, and is therefore a single point of failure. One immediately replaces clients having one key by clients having multiple keys. This makes it more secret since a big spender can hide his/her identify behind many small accounts (Fig. 1 (b)).

The ledger can be derived from the transactions and the amount associated with each key at time 0, and therefore only a transaction log is needed (Fig. 2 (c)).

### 3.2 Refining the Model – Block Rewards, Transaction Fees, Signatures

We will pause with the exposition by Warner. However, we will already at this stage add as well the notions of a block, and block rewards to the model, and discuss signatures and addresses.

The proof of work (“miners’ riddle”) will be added to the model in a followup article. We will here model the reward given to the miner. There is a special transaction, called coinbase transaction, which has no inputs, and where the miner decides about the outputs. The sum of its outputs needs to be equal to the sum of the block reward and the transaction fees. Transaction fees are the difference between the amount in the inputs and the outputs of previous transactions.

At the time of writing, in order for a transaction to be accepted in the next blockchain, a transaction fee needs to be added since there is not enough space in a block to cover all transactions. The block reward is determined by a function from time (which measures the number of blocks) to the amounts. This function is for Bitcoin the function which starts at time zero with 50 bitcoins per block, halves every 52,500 blocks, and goes to zero from block 6,930,000 onwards.

An address is the hash of a public key. In order to check whether a message was signed, one needs therefore to provide the public key, which hashes to the address in question, and check that the message was signed by the private key associated with the public key (see the discussion that hash actually means double hashing in Subsect. 3.3 below).

We finally discuss the signatures needed for transactions. For each input, what needs to be signed is not only that the originator spends the amount in the given transaction. Otherwise, a hacker could easily make a replay attack: if after a transaction there is still enough money in the account, one can create another transaction with the same input and signed by the same signature, which then goes to a different output. To avoid this, what needs to be signed is a message consisting of both the input and all outputs. In the Bitcoin protocol, the sender signs only his own input together with the outputs, not the other inputs. This way of signing doesn’t prevent a replay attack, as we will see in Sect.5. There we develop the transaction based tree model which models more accurately the Bitcoin model and prevents this form of replay attack.

### 3.3 Cryptographic Assumptions

We will in this paper not implement cryptography, but just postulate certain functions. We define messages, which can be formed from natural numbers by using the sum of two messages and list of messages. We assume a hash function which creates from a message, which can be encoded as a natural number, the hash. (In the real Bitcoin protocol, different hash functions are applied for different purposes, and usually two hash functions are applied. By hashing we mean in the following the application of one or two hash functions, as it is used in the Bitcoin protocol for that instance). When proving properties in later papers, we assume this function to be injective. This is not really the case for the real hash functions, but it is assumed that the probability of having having a clash is very low. We assume as well a hash function (again more precisely double-hash function) from public keys to addresses, and a predicate which checks for a message, a public key, and a signature, whether the message was signed by the private key corresponding to this public key.

## 4 The Bitcoin Ledger in Agda

We will in the following develop the model from Sect. 3 in Agda. We start by introducing basic data types defining the `Time` (which is the number of blocks taken up to now), `Amount` (amount of bitcoins being used), `Address`, transaction ids `TXID`, signatures `Signature`, public keys `PublicKey`. As common in the literature, we use TX for transactions. All of these types will be equated to the set of natural numbers (we define here only the first one):

```
Time = ℕ
```

We define a data type of messages, together with a hash function which maps messages to natural numbers. A message can be a natural number (which might be a time, amount, address etc.), it may consist of two messages, or it is a list of messages. Using this we can then form later a message encoding for instance a transaction which consists of a list of inputs and outputs, each of which consists of several components. The set of messages together with a postulated hash function for messages is given as follows:

```
data Msg : Set where
  nat      : (n : ℕ) → Msg
  _+msg_   : (m m' : Msg) → Msg
  list     : (l : List Msg) → Msg
```

```
postulate hashMsg : Msg → ℕ
```

We postulate a function which converts public keys into addresses (which corresponds in Bitcoin to another application of hash functions):

```
postulate publicKey2Address : (pubk : PublicKey) → Address
```

We assume a predicate `Signed` which determines, whether a message was signed by the private key corresponding to a given public key. We introduce as well a set `SignedWithSigPbk`, consisting of a signature, a public key, the property that a message was signed by it, and that the address is the hashed version of the public key. `SignedWithSigPbk` is an example of a record type which depends on parameters (here `msg`, `publicKey` and `s`):

```
postulate Signed : (msg : Msg)(publicKey : PublicKey)(s : Signature) → Set
```

```
record SignedWithSigPbk (msg : Msg)(address : Address) : Set where
  publicKey : PublicKey
  pbkCorrect : publicKey2Address publicKey ≡ℕ address
  signature  : Signature
  signed     : Signed msg publicKey signature
```

An input or output of a transaction is given by the amount being received or sent, and the address of the sender or recipient:

```
record TXField : Set where
  constructor txField
  amount : Amount
  address : Address
```

We create a message for a field consisting of the amount and the address, and extend it to a list of fields (`mapL` is the function applying a function to each element in a list):

```
txField2Msg : (inp : TXField) → Msg
txField2Msg inp = nat (amount inp) +msg nat (address inp)
```

```
txFieldList2Msg : (inp : List TXField) → Msg
txFieldList2Msg inp = list (mapL txField2Msg inp)
```

The sum of inputs and outputs is computed using the following function

```
txFieldList2TotalAmount : (inp : List TXField) → Amount
txFieldList2TotalAmount inp = sumListViaf amount inp
```

Here we use the function `sumListViaf` which sums up the result of applying a function to each element of a given list:

```
sumListViaf : {X : Set} (f : X → ℕ)(l : List X) → ℕ
sumListViaf f [] = 0
sumListViaf f (x :: l) = f x + sumListViaf f l
```

A transaction without signature consists of a list of inputs and outputs:

```
record TXUnsigned : Set where
  inputs  : List TXField
  outputs : List TXField
```

We compute a message for it

```
txUnsigned2Msg : (transac : TXUnsigned) → Msg
txUnsigned2Msg transac = txFieldList2Msg (inputs transac) +msg txFieldList2Msg (outputs transac)
```

As discussed in Sect. 5, the message to be signed for an input consists of the input and the list of outputs:

```
txInput2Msg : (inp : TXField)(outp : List TXField) → Msg
txInput2Msg inp outp = txField2Msg inp +msg txFieldList2Msg outp
```

The set of signatures for a transaction consists of a signature for each input using the function `txInput2Msg`. Note that one needs to provide the public key, which hashes to the given address together with a signature for the message created:

```
tx2Signaux : (inp : List TXField) (outp : List TXField) → Set
tx2Signaux [] outp = T
tx2Signaux (inp :: restinp) outp =
  SignedWithSigPbk (txInput2Msg inp outp) (address inp) × tx2Signaux restinp outp
```

```
tx2Sign : TXUnsigned → Set
tx2Sign tr = tx2Signaux (inputs tr) (outputs tr)
```

A transaction is an unsigned transaction, for which the sum of outputs is greater or equal than the sum of inputs, the list of inputs and outputs is non-empty, and which is signed by the private key of the public keys, which in turn hashes to the input addresses:

```
record TX : Set where
  tx      : TXUnsigned
```

```

cor      : txFieldList2TotalAmount (inputs tx) ≥ txFieldList2TotalAmount (outputs tx)
nonEmpty : NonNil (inputs tx) × NonNil (outputs tx)
sig      : tx2Sign tx

```

A ledger gives for every time and address the amount associated with that address. A ledger row (for brevity we use in Agda ledger instead of ledger row) is therefore a function from addresses to amount. We define as well the empty ledger where the amount is 0:

```
Ledger = (addr : Address) → Amount
```

```
initialLedger : Ledger
```

```
initialLedger addr = 0
```

We define the result of updating a ledger by adding the values in a transaction, i.e. the amount for the person given by the address in the field is incremented by the amount in the field. This operation is then extended to a list of transaction:

```
addTXFieldToLedger : (tr : TXField)(oldLedger : Ledger) → Ledger
```

```
addTXFieldToLedger tr oldLedger pubk =
```

```
  if pubk ≡ ℕb address tr then oldLedger pubk + amount tr else oldLedger pubk
```

```
addTXFieldListToLedger : (tr : List TXField)(oldLedger : Ledger) → Ledger
```

```
addTXFieldListToLedger []      oldLedger = oldLedger
```

```
addTXFieldListToLedger (x :: tr) oldLedger =
```

```
  addTXFieldListToLedger tr (addTXFieldToLedger x oldLedger)
```

The result of subtracting a ledger field is defined in the same way by replacing  $+$  by  $\dot{-}$ . Here  $\dot{-}$  is the cut off subtraction for natural numbers, replacing negative results of subtraction by zero, e.g.  $3 \dot{-} 2 = 1$  and  $2 \dot{-} 3 = 0$ . We give here only the types of the functions:

```
subtrTXFieldFromLedger : (tr : TXField) (oldLedger : Ledger) → Ledger
```

```
subtrTXFieldListFromLedger : (tr : List TXField) (oldLedger : Ledger) → Ledger
```

A ledger is updated by a transaction by first subtracting the inputs and then adding the outputs:

```
updateLedgerByTX : (tr : TX)(oldLedger : Ledger) → Ledger
```

```
updateLedgerByTX tr oldLedger = addTXFieldListToLedger (outputs (tx tr))
```

```
  (subtrTXFieldListFromLedger (inputs (tx tr)) oldLedger)
```

An input is correct of a ledger is correct w.r.t. a ledger row, if there is enough money in the ledger to pay for it:

```
correctInput : (tr : TXField) (ledger : Ledger) → Set
```

```
correctInput tr ledger = ledger (address tr) ≥ amount tr
```

A list of inputs of a ledger is correct if each input is correct, while updating the ledger after each input, since there might be several inputs for the same address. This function is then extended to a transaction:

```
correctInputs : (tr : List TXField) (ledger : Ledger) → Set
```

```
correctInputs []      ledger = ⊤
```

$$\text{correctInputs } (x :: tr) \text{ ledger} = \text{correctInput } x \text{ ledger} \times \\ \text{correctInputs } tr \text{ (subtrTXFieldFromLedger } x \text{ ledger)}$$

$$\text{correctTX} : (tr : TX) (ledger : Ledger) \rightarrow \text{Set}$$

$$\text{correctTX } tr \text{ ledger} = \text{correctInputs } (\text{outputs } (tx \ tr)) \text{ ledger}$$

An unmined block (which means there is no correctness check) is a list of transactions:

$$\text{UnMinedBlock} = \text{List TX}$$

The transaction fee of a transaction is the difference between the sum of the inputs and sum of the outputs of the transaction. The transaction fee of an unmined block is the sum of transaction fees of its transactions.

$$\text{tx2TXFee} : TX \rightarrow \text{Amount}$$

$$\text{tx2TXFee } tr = \\ \text{txFieldList2TotalAmount } (\text{outputs } (tx \ tr)) \dot{-} \text{txFieldList2TotalAmount } (\text{inputs } (tx \ tr))$$

$$\text{unMinedBlock2TXFee} : \text{UnMinedBlock} \rightarrow \text{Amount}$$

$$\text{unMinedBlock2TXFee } bl = \text{sumListViaf } \text{tx2TXFee } bl$$

A block is correct if each transaction is correct, while updating the ledger after each transaction. Note that Bitcoin allows to have in one block transactions which make use of the output of another transaction [10]. We determine as well the result of updating the ledger after a block:

$$\text{correctUnminedBlock} : (block : \text{UnMinedBlock})(oldLedger : Ledger) \rightarrow \text{Set}$$

$$\text{correctUnminedBlock } [] \quad \quad \quad oldLedger = \top$$

$$\text{correctUnminedBlock } (tr :: block) \text{ oldLedger} = \\ \text{correctTX } tr \text{ oldLedger} \times \text{correctUnminedBlock } block \text{ (updateLedgerByTX } tr \text{ oldLedger)}$$

$$\text{updateLedgerUnminedBlock} : (block : \text{UnMinedBlock})(oldLedger : Ledger) \rightarrow \text{Ledger}$$

$$\text{updateLedgerUnminedBlock } [] \quad \quad \quad oldLedger = oldLedger$$

$$\text{updateLedgerUnminedBlock } (tr :: block) \text{ oldLedger} = \\ \text{updateLedgerUnminedBlock } block \text{ (updateLedgerByTX } tr \text{ oldLedger)}$$

An unchecked block consists of an unmined block and the output given to the miner. The miner adds one transaction to a block, called coinbase transaction, which has no inputs, only outputs, such that the sum of outputs is equal to the sum of the block reward and the transaction fees of that block. We compute as well the transaction fee of a block:

$$\text{BlockUnchecked} = \text{List TXField} \times \text{UnMinedBlock}$$

$$\text{block2TXFee} : \text{BlockUnchecked} \rightarrow \text{Amount}$$

$$\text{block2TXFee } (outputMiner, block) = \text{unMinedBlock2TXFee } block$$

A block is correct, if the unmined block is correct and the output of the coinbase transaction is equal to the sum of the block reward and the transaction fees:

$$\text{correctMinedBlock} : (reward : \text{Amount})(block : \text{BlockUnchecked})(oldLedger : Ledger) \rightarrow \text{Set}$$

```

correctMinedBlock reward (outputMiner , block) oldLedger =
  correctUnminedBlock block oldLedger ×
  txFieldList2TotalAmount outputMiner ≡ℕ (reward + unMinedBlock2TXFee block)

```

After a block the ledger is updated by updating it using the unmined block and adding the output of the coinbase transaction:

```

updateLedgerBlock : (block : BlockUnchecked)(oldLedger : Ledger)→ Ledger
updateLedgerBlock (outputMiner , block) oldLedger =
  addTXFieldListToLedger outputMiner (updateLedgerUnminedBlock block oldLedger)

```

An unchecked blockchain is a list of unchecked blocks:

```
BlockChainUnchecked = List BlockUnchecked
```

The correctness of a block depends on the block reward which is determined by a block reward function, which determines the reward depending on time (which is the number of blocks). Assuming a start time and an initial ledger (which for the final blockchain will be set to 0 and `initialLedger`), the correctness of a blockchain is defined as follows:

```

CorrectBlockChain : (blockReward : Time → Amount)
                   (startTime : Time)
                   (startLedger : Ledger)
                   (bc : BlockChainUnchecked)
                   → Set
CorrectBlockChain blockReward startTime startLedger [] = T
CorrectBlockChain blockReward startTime startLedger (block :: restbc)
  = correctMinedBlock (blockReward startTime) block startLedger
    × CorrectBlockChain blockReward (suc startTime)
      (updateLedgerBlock block startLedger) restbc

```

The ledger at the end of the blockchain is defined as follows:

```

FinalLedger : (txFeePrevious : Amount) (blockReward : Time → Amount)
              (startTime : Time)      (startLedger : Ledger)
              (bc : BlockChainUnchecked) → Ledger
FinalLedger trfee blockReward startTime startLedger [] = startLedger
FinalLedger trfee blockReward startTime startLedger (block :: restbc) =
  FinalLedger (block2TXFee block) blockReward (suc startTime)
  (updateLedgerBlock block startLedger) restbc

```

A blockchain is an unchecked blockchain which is correct :

```

record Blockchain (blockReward : Time → Amount) : Set where
  blockchain : BlockChainUnchecked
  correct    : CorrectBlockChain blockReward 0 initialLedger blockchain

```

We determine the ledger at the end of a blockchain:

```

blockChain2FinalLedger : (blockReward : Time → Amount)(bc : Blockchain blockReward)
                       → Ledger
blockChain2FinalLedger blockReward bc =
  FinalLedger 0 blockReward 0 initialLedger (blockchain bc)

```

**5 From a Ledger to a Transaction Tree**

The problem of the ledger is that transactions are not unique. If for instance a user (as given by an address) has 10 bitcoins, and transfers 4, somebody can make a replay attack and repeat the same transaction, deducting 8 bitcoins from that account. The data for both transactions is the same, so the message and signature will be the same.

One might think that one could prevent this by adding for instance the time to the message of a transaction. Then transactions in different blocks would be different, and therefore their messages and signatures would be different, and a replay attack is not possible in a different block. However, Bitcoin allows transactions which make use of the output of a transaction in the same block [10], and therefore transactions using the same input address should be allowed as well. Transactions in the same block cannot be distinguished by their block number; they could only distinguished by their physical time, which is difficult to verify in a peer-to-peer network.

Therefore in Bitcoin a different approach has been taken: a transaction doesn't refer just to an address but it refers to the unspent output of a previous transaction. Should transactions be unique, this would make a replay attack impossible. It turns out that this was originally not the case in the Bitcoin protocol but has been fixed since, see the discussion in Sect. 7. Once transactions are unique, and one cannot infer from a signature of one transaction the signature of another one, replay attacks are no longer possible.

An example is shown in Fig. 2 (a). In that diagram we have to the left a coinbase transaction with outputs 1, 6, 4. Here the sum of outputs are equal to the input, which is the sum of previous transaction fees and block reward. All other transactions have a transaction fee of 1. UTXO are the unspent transaction outputs, which can be used in future transactions.




Fig. 2 (a) Transaction Tree




Fig. 2 (b) Serialised Transaction  
Source: [33] (License: as Fig. 1)

It follows that one needs to define the transactions, the transaction tree, and the unspent transactions simultaneously. In order to determine the inputs for coinbase transactions one needs as well to compute the transaction fees. This gives rise to an inductive-recursive [14, 16, 15] definition (in an extended sense): one defines inductively the set of transactions and trees formed from transactions, while recursively simultaneously defining the list of unspent transaction outputs.<sup>3</sup> In order to determine the inputs for coinbase transactions,

<sup>3</sup> This is however not a standard form of induction-recursion, since the unspent transactions are lists of transactions, which means they refer to a set one is defining inductively simultaneously. It could

one needs to compute as well the set of transaction fees.

When broadcasting transactions, one needs to reference previous transaction outputs. This is done using Merkle trees [24], see Fig 2 (b): One computes, by recursion over the transaction tree, for each transaction an id. The id for a transaction is the hashed message obtained as follows: for each input one takes the transaction id of the transaction used together with the number of the output used, and the signature. The outputs consist of the amount and the address of the recipient. All of this is then hashed, resulting in the transaction id of the transaction, which can be used in future transactions.

One last detail to be considered is the maturation time. Mining is used in order to find a consensus which block chain is the correct one. We will not elaborate on this consensus protocol here, since we have still to include mining into our model of Bitcoin. A malicious miner can create a false blockchain, but that chain is expected to die out fast, because of the consensus protocol. However, for a few blocks it might still be active. In order to avoid that people spent money from malicious miners, coinbase transactions can only be used after a certain number of blocks (currently 100) has been mined. This number is called the maturation time, which we will take into account in our model.

## 6 Transaction Trees in Agda

### 6.1 Preliminaries

The basic set up such as `Time`, `Amount`, etc, `Msg`, `hashMsg`, `publicKey2Address`, `Signed` are as before. We will in this section fix the `minerReward`. We define as well `blockMaturationTime`, which determines the time after which the reward obtained by miners can be used. In Bitcoin the value is currently 100.

```
postulate blockReward : (t : Time) → Amount
postulate blockMaturationTime : Time
```

The output field of a transaction is the same as the `TXField` in Sect. 6:

```
record TXOutputfield : Set where
  constructor txOutputfield
  amount : Amount
  address : Address
```

### 6.2 Inductive-Recursive Definition of Transaction Tree, Transactions, and Unspent Transaction Outputs

We are going to give the inductive-recursive definition (in an extended sense) of the transactions and unspent transaction outputs. This is a relatively long simultaneous definition: We define inductively the set of transaction trees `TXTree`, the transactions `TX`, and the set of transaction outputs `TXOutput`, while recursively defining the unspent transaction outputs `utxo`. In addition we define simultaneously recursively auxiliary functions, determining the

---

be reduced to a simultaneous inductive definition by defining first transaction trees, transactions, and transaction outputs simultaneously inductively, without guaranteeing that only unspent transaction outputs are used as inputs. Then one could select those transaction trees, in which only unspent transaction outputs are used. However, that would make the model much more difficult to understand and reason about.

type of possible transaction inputs `TxInputs` for a transaction, the set of transaction outputs `tx2TXOutputs` of a transaction, and the number of outputs `nrOutputs` of a transaction.

The set `TXTree` is essentially a sequence of transactions, however the inputs of a transaction refer to previous transactions in that sequence. The tree `genesisTree` is the starting point of a blockchain, where there are no transactions yet:

```
data TXTree : Set where
  genesisTree : TXTree
  txtree      : (tree : TXTree)(tx : TX tree) → TXTree
```

A transaction can be a normal transaction with inputs being previous transactions, and a coinbase transaction, where there are no inputs, but we annotate the time (i.e. the block number to which it belongs):

```
data TX (tr : TXTree) : Set where
  normalTX : (inputs : TxInputs tr) (outputs : List TXOutputfield) → TX tr
  coinbase  : (time : Time)          (outputs : List TXOutputfield) → TX tr
```

We will record the set of unspent transaction outputs, where a transaction output is given by a transaction belonging to a tree, and an output number in that tree (since it occurs as part of a simultaneous inductive definition, it is marked by the keyword `inductive`):

```
record TXOutput : Set where
  inductive
  constructor txOutput
  trTree : TXTree
  tx      : TX trTree
  output : Fin (nrOutputs trTree tx)
```

Simultaneously we define recursively the set of unspent transactions in a transaction tree, where we give the type now, but the definition later. We define as well as an auxiliary definition `utxoMinusNewInputs` which determines the result of deleting the inputs of a transaction from the list of unspent transaction.

```
utxoMinusNewInputs : (tr : TXTree)(tx : TX tr) → List TXOutput
utxo : (tr : TXTree) → List TXOutput
```

The set of inputs of a transaction will be a sublist of the set of unspent transaction outputs. A sublist of a list is given by a sequence of indices from the original list, but after each index deleting that element from that list:

```
data SubList {X : Set} : (l : List X) → Set where
  [] : {l : List X} → SubList l
  cons : {l : List X}(i : Fin (length l))(o : SubList (delFromList l i)) → SubList l
```

We define result of deleting a sublist from a list and a function converting a sublist into the underlying list:

```
listMinusSubList : {X : Set}(l : List X)(o : SubList l) → List X
listMinusSubList l [] = l
listMinusSubList l (cons i o) = listMinusSubList (delFromList l i) o
```

```

subList2List : {X : Set}{l : List X}(sl : SubList l) → List X
subList2List [] = []
subList2List {l = l'} (cons i sl) = nth l i :: subList2List sl

```

The set of inputs will be a sublist of the list of unspent transaction, but we need to add as well the public keys and signatures used. We don't add any checking of the correctness of the signature – this will be defined later as an extra predicate. We first introduce the notion **SubList+**, which is a sublist, with some extra information (of type  $Y$ ) added for each element:

```

data SubList+ {X : Set} (Y : Set) : (l : List X) → Set where
[] : {l : List X} → SubList+ Y l
cons : {l : List X}{i : Fin (length l)}(y : Y)(o : SubList+ Y (delFromList l i))
      → SubList+ Y l

```

The subtraction of **SubList+** from a list is defined as before, while ignoring the extra elements:

```

listMinusSubList+ : {X Y : Set}(l : List X)(o : SubList+ Y l) → List X

```

The underlying list of a sublist contains as well the extra elements of the sublist, defined similarly to before, having type:

```

subList+2List : {X Y : Set}{l : List X}(sl : SubList+ Y l) → List (X × Y)

```

We define as well two operations on sublists needed later. One is **listMinusSubList+Index2OrgIndex** mapping indices from the list after subtracting from it a sublist to indices of the original list:

```

listMinusSubList+Index2OrgIndex : {X Y : Set}(l : List X)(o : SubList+ Y l)
                                   (i : Fin (length (listMinusSubList+ l o))) → Fin (length l)

```

The other function takes a sublist and gives a list of all the elements of the sublist together with their indices in the original list:

```

subList+2IndicesOriginalList : {X Y : Set}(l : List X)(sl : SubList+ Y l) → List (Fin (length l) × Y)

```

With these definitions we can define the set of inputs as a sublist of the set of unspent outputs, but adding a signature and a public key:

```

TxInputs : (tr : TXTree) → Set
TxInputs tr = SubList+ (PublicKey × Signature) (utxo tr)

```

The function **utxoMinusNewInputs** deletes, in case of a normal transaction, the inputs from the previous list of unspent transaction outputs, using the function **listMinusSubList+**. In case of a coinbase transaction, nothing needs to be deleted:

```

utxoMinusNewInputs tr (normalTX inputs outputs) = listMinusSubList+ (utxo tr) inputs
utxoMinusNewInputs tr (coinbase time outputs) = utxo tr

```

The list of new outputs of a transaction to be added to **utxo** is a list of pairs consisting of the new transaction and an index. The indices are all elements of **Fin n** where  $n$  is the number of outputs of the transaction. We first define the function **listOfElementsOffFin n**, which returns a list containing all elements of **Fin n**:

```
listOfElementsOfFin : (n : ℕ) → List (Fin n)
listOfElementsOfFin zero = []
listOfElementsOfFin (suc n) = zero :: (mapL suc (listOfElementsOfFin n))
```

The list of outputs is obtained by mapping the elements of this list to the corresponding output. In the following  $(\lambda i \rightarrow \text{txOutput } tr \ tx \ i)$  denotes the definition of an anonymous function, mapping  $i$  to  $\text{txOutput } tr \ tx \ i$ :

```
tx2TXOutputs : (tr : TXTree)(tx : TX tr) → List TXOutput
tx2TXOutputs tr tx = mapL (\ i → txOutput tr tx i)(listOfElementsOfFin (nrOutputs tr tx))
```

The set of unspent transaction outputs is now defined as follows: For `genesisTree` it is empty. If we add a transaction to the transaction tree, we first delete the inputs spent in it and add its outputs:

```
utxo genesisTree = []
utxo (txtree tr tx) = utxoMinusNewInputs tr tx ++ tx2TXOutputs tr tx
```

Finally, `nrOutputs` determines the number of outputs of a transaction, which is the length of the list of outputs:

```
nrOutputs : (tr : TXTree) (tx : TX tr) → ℕ
```

This concludes the inductive-recursive definitions of transaction trees, transactions and unspent transactions.

### 6.3 Computing Sum of Inputs and Outputs of Transactions and Transaction Fees

We need to define what it means that the sum of inputs of transaction is greater or equal than the sum of its outputs. The sum of outputs is computed as follows:

```
outputs2SumAmount : List TXOutputfield → Amount
outputs2SumAmount l = sumListViaf amount l

tx2SumOutputs : {tr : TXTree}(tx : TX tr) → Amount
tx2SumOutputs (normalTX inputs outputs) = outputs2SumAmount outputs
tx2SumOutputs (coinbase time outputs) = outputs2SumAmount outputs
```

The inputs of a transaction can be mapped to a list of `TXOutput`, which consists of a transaction and an output number. This can be mapped to a `TXOutputfield`, which determines the amount being output and the address of the recipient. The following functions extract this information from an element of `TXOutput`:

```
txOutput2Outputfield : TXOutput → TXOutputfield
txOutput2Outputfield (txOutput trTree (normalTX inputs outputs) i) = nth outputs i
txOutput2Outputfield (txOutput trTree (coinbase time outputs) i) = nth outputs i

txOutput2Amount : TXOutput → Amount
txOutput2Amount output = txOutput2Outputfield output .amount
```

```
txOutput2Address : TXOutput → Address
txOutput2Address output = txOutput2Outputfield output .address
```

Now we lift the inputs of transaction to a list consisting of a `TXOutput`, a public key, and a signature. We compute as well the sum of outputs of a transaction:

```
inputs2PrevOutputsSigPbk : (tr : TXTree)(inputs : TxInputs tr)
  → List (TXOutput × PublicKey × Signature)
inputs2PrevOutputsSigPbk tr inputs = subList+2List inputs
```

```
inputs2PrevOutputs : (tr : TXTree)(inputs : TxInputs tr) → List TXOutput
inputs2PrevOutputs tr inputs = mapL proj₁ (inputs2PrevOutputsSigPbk tr inputs)
```

```
inputs2Sum : (tr : TXTree)(inputs : TxInputs tr) → Amount
inputs2Sum tr inputs = sumListViaf txOutput2Amount (inputs2PrevOutputs tr inputs)
```

The above will compute the sum of inputs for a normal transaction. For coinbase transaction there is no explicit input, but the output needs to be equal to the transaction fees and the reward for the current block. So we need to compute the time of the current block, which determines the block reward. The time for the current block is updated whenever we have a coinbase transaction. (The correctness condition for a `TXtree` will guarantee that the times in coinbase transactions are chosen correctly):

```
txTree2TimeNextTobeMinedBlock : (tr : TXTree) → Time
txTree2TimeNextTobeMinedBlock genesisTree = 0
txTree2TimeNextTobeMinedBlock (txtree tree (normalTX inputs outputs)) =
  txTree2TimeNextTobeMinedBlock tree
txTree2TimeNextTobeMinedBlock (txtree tree (coinbase time outputs)) = suc time
```

Now we compute for simultaneously recursively the sum of inputs of a transaction (which is in case of a coinbase transaction the sum of transaction fees and the block reward) and the transaction fees in the next block to be mined (which is reset after a coinbase transaction to zero):

```
mutual
tx2SumInputs : (tr : TXTree)(tx : TX tr) → Amount
tx2SumInputs tr (normalTX inputs outputs) = inputs2Sum tr inputs
tx2SumInputs tr (coinbase time outputs) =
  txTree2RecentTXFees tr + blockReward (txTree2TimeNextTobeMinedBlock tr)
```

```
txTree2RecentTXFees : (tr : TXTree) → Amount
txTree2RecentTXFees genesisTree = 0
txTree2RecentTXFees (txtree tr (normalTX inputs outputs)) =
  txTree2RecentTXFees tr + (inputs2Sum tr inputs ÷ outputs2SumAmount outputs)
txTree2RecentTXFees (txtree tr (coinbase time outputs)) = 0
```

Coin transactions can only be used in a transaction after the maturation time has passed. We therefore define the maturation time of an output of a transaction:

```
output2MaturationTime : TXOutput → Time
output2MaturationTime (txOutput trTree (normalTX inputs outputs) i) = 0
output2MaturationTime (txOutput trTree (coinbase time outputs) i) = time + blockMaturationTime
```

## 6.4 Computing Messages to be Signed and Transaction IDs

We will compute the messages to be signed when an unspent transaction output is used in a transaction, and the messages and corresponding transaction ids for transactions. We first compute the messages for an output field and a list of outputs of a transaction, which is a list of output fields:

```

txOutputfield2Msg : TXOutputfield → Msg
txOutputfield2Msg (txOutputfield amount1 address1) = nat amount1 +msg nat address1

outputFields2Msg : (outp : List TXOutputfield) → Msg
outputFields2Msg outp = list (mapL txOutputfield2Msg outp)

```

We will now define simultaneously the messages and ids of transactions, and the messages for unspent transactions. The latter is needed to construct the message for a transaction, since the inputs are using unspent transactions. One might think that one can compute the messages for unspent transaction outputs directly from previous transaction ids. However, such a definition does not terminate in Agda, since there is no structural reason that those previous transaction ids were computed before the current transaction. By defining simultaneously with the transaction messages and ids the messages for unspent transaction, we overcome this problem. Later, one can show that the messages of the unspent transactions are indeed equal to the combination of the transaction id and output number.

The message for a normal transaction consists of the message for the outputs, as defined before, and the message for the inputs. The message for the inputs is computed by first computing indices in the list of unspent transactions together with public keys and signatures, obtained from the inputs. Then we construct from each input the message consisting of the message for the corresponding unspent transaction, the public key and the signature. Note that at this moment we haven't defined what it means for a transaction to be correctly signed.

For coinbase transactions the message consists of the time and the message for the outputs. The fact that we include the time is a fix taken in Bitcoin to guarantee that transactions are unique. This will be discussed in Subsect. 7.2. The computation of the message for a transaction is as follows (the part starting with `where` starts auxiliary local definitions):

```

tx2Msg : (tr : TXTree)(tx : TX tr) → Msg
tx2Msg tr (normalTX inputs1 outputs1) = list (mapL utxoIndexSig2Msg inputIndices)
                                         +msg outputFields2Msg outputs1

where
  inputIndices : List (Fin (length (utxo tr)) × PublicKey × Signature)
  inputIndices = subList+2IndicesOriginalList (utxo tr) inputs1

  utxoIndexSig2Msg : Fin (length (utxo tr)) × PublicKey × Signature → Msg
  utxoIndexSig2Msg (i , (pbk , sig)) = utxo2Msg tr i +msg nat pbk +msg nat sig

tx2Msg tr (coinbase time outputs1) = nat time +msg outputFields2Msg outputs1

```

We obtain the id of a transaction:

```

tx2id : (tr : TXTree)(tx : TX tr) → ℕ
tx2id tr tx = hashMsg (tx2Msg tr tx)

```

In order to compute the messages for the unspent transactions, we compute first the messages for the unspent transactions of the previous transaction tree after deleting the inputs for the new transaction:

```

utxoMinusNewInputs2Msg : (tr : TXTree)(tx : TX tr)(i : Fin (length (utxoMinusNewInputs tr tx)))
  → Msg
utxoMinusNewInputs2Msg tr (normalTX inputs outputs) i =
  utxo2Msg tr (listMinusSubList+Index2OrgIndex (utxo tr) inputs i)
utxoMinusNewInputs2Msg tr (coinbase time outputs) i = utxo2Msg tr i

```

The new list of unspent transaction is the concatenation of two lists. We define a function which, if one has two lists and functions mapping the indices of those two lists to the elements in those lists, computes the indexing function which does the same for the concatenation of both lists. It has type

```

concatListIndex2OriginIndices : {X Y : Set}(l l' : List X)
  (f : Fin (length l) → Y)(f' : Fin (length l') → Y)
  (i : Fin (length (l ++ l'))) → Y

```

We use this function for calculating the messages for unspent transaction by calculating these functions for the result of omitting the inputs from the unspent transactions and for the new outputs of a transaction. In the following definition there occur two more Agda features not discussed before: The occurrence of `()` after `genesisTree` denotes the empty case distinction. `genesisTree` has no unspent transactions, therefore `(utxo genesisTree)` is empty, and `(Fin (length (utxo genesisTree)))` is the empty set. So the case distinction on this set is empty, and denoted by `()`. Furthermore, we have an occurrence of `module utxo2Msgaux where`, which is like a `where` clause, but allows to reference the local variables by using for instance `utxo2Msgaux.f1` for denoting `f1`. The definition of `utxo2Msg` is as follows:

```

utxo2Msg : (tr : TXTree)(i : Fin (length (utxo tr))) → Msg
utxo2Msg genesisTree ()
utxo2Msg (txtree tr tx) = concatListIndex2OriginIndices l0 l1 f0 f1
module utxo2Msgaux where
  l0 : List TXOutput
  l0 = utxoMinusNewInputs tr tx

  l1 : List TXOutput
  l1 = tx2TXOutputs tr tx

  f0 : Fin (length l0) → Msg
  f0 i = utxoMinusNewInputs2Msg tr tx i

  f1 : Fin (length l1) → Msg
  f1 i = nat (tx2id tr tx) +msg nat (toN i)

```

The message to be signed by the owner of the address in a transaction output consists of the input of the transaction and all the outputs of the transaction. Interestingly, in Bitcoin only the current input is used, not the other ones. Furthermore, for the input we cannot include the signature in the message to be signed, because that signature is created from the message to be signed. So the input to be signed consists of the transaction id for that output, the output number, and the address in that output:

```

msgToBeSignedByInput : (txoutput : TXOutput)(outputs : List TXOutputfield) → Msg
msgToBeSignedByInput txoutput outputs =
  (nat (tx2id (trTree txoutput) (tx txoutput)) +msg
   nat (toN (output txoutput)) +msg
   nat (txOutput2Address txoutput)) +msg
  outputFields2Msg outputs

```

In order to define the correctness of a tree, we define first an operation expressing that for all elements of a list a given property P holds:

```

∀inList : {X : Set}(l : List X)(P : X → Set) → Set
∀inList [] P      = ⊤
∀inList (x :: l) P = P x × ∀inList l P

```

Now we define the correctness of Transaction. A normal transaction is correct, if the inputs and outputs are non-empty, the sum of inputs is greater or equal to the sum of outputs, for all inputs used the maturation time has passed, and for all inputs the public key given hashes to the address of the unspent transaction output being used, and the signature given is actually a signature of the message to be signed for that input. A coinbase transaction is correct, if the outputs are non empty, the sum of outputs is equal to the transaction fees obtained and the block reward, and the time is the time of the currently to be mined block. In the following definition  $\lambda\{(outp, pbk, sign) \rightarrow \dots\}$  denotes a local function definition, where we pattern match on the argument, obtaining that it is of the form  $(outp, pbk, sign)$ .

```

CorrectTX : (tr : TXTree)(tx : TX tr) → Set
CorrectTX tr (normalTX inputs outputs) =
  NonNil (inputs2PrevOutputs tr inputs) ×
  NonNil outputs ×
  (inputs2Sum tr inputs ≥ outputs2SumAmount outputs) ×
  (∀inList (inputs2PrevOutputs tr inputs)
   (λ o → output2MaturationTime o ≤ txTree2TimeNextTobeMinedBlock tr)) ×
  (∀inList (inputs2PrevOutputsSigPbk tr inputs)
   (λ {(outp, pbk, sign) →
     publicKey2Address pbk ≡ txOutput2Address outp ×
     Signed (msgToBeSignedByInput outp outputs) pbk sign })))

CorrectTX tr (coinbase time outputs) =
  NonNil outputs ×
  outputs2SumAmount outputs ≡ txTree2RecentTXFees tr + blockReward time ×
  time ≡ txTree2TimeNextTobeMinedBlock tr

```

A transaction tree is correct if all its transactions are correct, and we define as well the notion of a correct transaction tree:

```

CorrectTxTree : (tr : TXTree) → Set
CorrectTxTree genesisTree = ⊤
CorrectTxTree (txtree tr tx) = CorrectTxTree tr × CorrectTX tr tx

record TXTreeCorrect : Set where

```

```

constructor txTreeCorrect
  txtr      : TXTree
  corTxtr   : CorrectTxTree txtr

```

We define as well the notion of a correct transaction

```

record TXcorrect (tr : TXTreeCorrect) : Set where
  theTx : TX (tr .txtr)
  corTx  : CorrectTX (tr .txtr) theTx

```

We define operations for defining the genesis tree as a correct transaction tree and adding a correct transaction to it. Alternatively, we could as well have defined `TXTreeCorrect` by having these operations as constructors together with functions mapping it to the underlying transaction trees and correctness proofs:

```

initialTxTreeCorrect : TXTreeCorrect
initialTxTreeCorrect = txTreeCorrect genesisTree tt

addTxTreeCorrect : (tr : TXTreeCorrect)(tx : TXcorrect tr) → TXTreeCorrect
addTxTreeCorrect tr tx =
  txTreeCorrect (txtree (tr .txtr) (tx .theTx)) (tr .corTxtr , tx .corTx)

```

## 6.5 Merkle Trees

A Bitcoin transaction is not referring directly to previous transactions, but to the ids of those transactions. This form of forming ids for the nodes of a tree, by determining for each node recursively first the hashes of the ids of its subtrees, and then hashing the resulting data in order to obtain an id for the node, is called a Merkle tree. We call transactions which refer to transaction ids in the following Merkle transactions. The outputs of a Merkle transaction are as before, but the inputs are elements of `MerkleInputField`:

```

record MerkleInputField : Set where
  constructor inputField
  txid      : TXID
  outputNr  : ℕ
  publicKey : PublicKey
  signature : Signature

```

Merkle transactions are coinbase transactions (which are the same as before) or standard transactions, where the inputs are elements of `MerkleInputField`:

```

record MerkleTXcoinbase : Set where
  constructor merkleCoinbase
  time      : Time
  outputs   : List TXOutputfield

record MerkleTXstandard : Set where
  constructor merkleTXstd
  inputs    : List MerkleInputField

```

```

    outputs : List TXOutputfield

data MerkleTX : Set where
  txcoinbase : MerkleTXcoinbase → MerkleTX
  txstd      : MerkleTXstandard → MerkleTX

```

We define a map from inputs of a transaction referring to a transaction tree to Merkle inputs:

```

txOutpSign2MerkleInput : TXOutput × PublicKey × Signature → MerkleInputField
txOutpSign2MerkleInput (txOutput tr tx1 i , pbk , sign) = inputField (tx2id tr tx1) (toℕ i) pbk sign

txInputs2MerkleInputs : (tr : TXTree)(txInputs : TxInputs tr) → List MerkleInputField
txInputs2MerkleInputs tr txin = mapL txOutpSign2MerkleInput (inputs2PrevOutputsSigPbk tr txin)

```

A Merkle transaction corresponds to a normal transaction, if the inputs and outputs are the same, modulo the maps defined before:

```

MerkleTXCorresponds2TX : (tx : MerkleTX)(tr : TXTree)(tx : TX tr) → Set
MerkleTXCorresponds2TX (txcoinbase (merkleCoinbase time1 outputs1)) tr (coinbase time2 outputs2)
  = time1 ≡ time2 × outputs1 ≡ outputs2
MerkleTXCorresponds2TX (txstd (merkleTXstd inputs1 outputs1)) tr (normalTX inputs2 outputs2)
  = inputs1 ≡ txInputs2MerkleInputs tr inputs2 × outputs1 ≡ outputs2
MerkleTXCorresponds2TX (txcoinbase x) tr (normalTX inputs1 outputs1) = ⊥
MerkleTXCorresponds2TX (txstd x) tr (coinbase time1 outputs1)      = ⊥

```

A correct Merkle transaction is a Merkle transaction which corresponds to a correct transaction:

```

record MerkleTXCor (tr : TXTreeCorrect) : Set where
  mk : MerkleTX
  mtX : TXcorrect tr
  cor : MerkleTXCorresponds2TX mk (tr .txtr) (mtX .theTx)

```

Finally we show that by adding correct Merkle transaction to a correct transaction tree we obtain a correct transaction tree:

```

updateTXTree : (tr : TXTreeCorrect)(m : MerkleTXCor tr) → TXTreeCorrect
updateTXTree tr m = addTxTreeCorrect tr (m .mtX)

```

What needs to be shown in the correctness proof, which will be presented in a follow up paper, is that for a correct Merkle tree, the message obtained from the Merkle tree is the same as the message of the corresponding TX, that this message is unique, and that we can actually decide whether a Merkle transaction is correct.

## 7 Correctness and Limitations of the Model

### 7.1 Limitations

While we have dealt with coinbase transaction, including transaction fees, block rewards, and maturation times, we haven't formalised mining yet. It should be relatively easy to

define the message of a block and define abstractly the proof of work (“the miner’s riddle”), and even the adjustment of difficulty should be relatively easy. However, formalising the consensus protocol, which guarantees that the chain with the main computational power wins, would be more challenging.

At the moment outputs are outputs to one single recipient. Bitcoin allows the outputs to be proper scripts. When using an output one needs to provide an input script which if concatenated with the output script results in the truth value true. Formalising the script language and adding input and output scripts to our model is probably not very difficult, since the script language of Bitcoin is very simple.

## 7.2 Correctness and Decidability

In this article we have provided a basic model of Bitcoin, but we haven’t proved any correctness or decidability properties.

If one starts with decidable cryptographic functions, all properties used in this article should be decidable. Proving it in Agda might be more challenging, especially in the Subject. 6.5, where one needs to determine for a Merkle transaction a corresponding normal transaction it belongs to – a pen and paper proof is relatively easy (check whether we have in the list of unspent transaction outputs those having the transaction id and output number needed).

Regarding correctness, it should be relatively easy to prove a simple property, namely that the sum of unspent transaction outputs is equal to the sum of the bitcoins mined. What is more challenging are the following properties:

- The signatures for transactions are unique. This requires the assumption that the hash functions are injective. Although this doesn’t hold, the probability of a clash is very low.
- If a transaction had an output, and this output is not used in a different transaction, it is in the unspent transaction list.
- If a transaction output has been spent, it cannot be used again.
- Any normal transaction input was the output of a transaction.

All these properties rely on the most important property:

- Prove that transactions messages and ids are unique.

All the above properties are provable on pen and paper. Proving them in Agda will take some time.

We want to note that originally in the Bitcoin protocol time (which corresponds to the block number) was not included in the message for coinbase transactions, and therefore identical coinbase transactions could be created, namely by having the same outputs. It actually occurred [28, 18]. The duplicate messages were the coinbase transaction in blocks 91842 and 91812<sup>4</sup> and as well in blocks 91880 and 91722<sup>5</sup> As a fix the height of the transaction was required to be added in the optional part of a coinbase transaction [17].

p

---

<sup>4</sup> The first (coinbase) transactions in the following blocks coincide:

<https://blockexplorer.com/block/00000000000743f190a18c5577a3c2d2a1f610ae9601ac046a38084ccb7cd721>  
<https://blockexplorer.com/block/00000000000271a2dc26e7667f8419f2e15416dc6955e5a6c6cdf3f2574dd08e>

<sup>5</sup> The coinbase transactions in the following blocks coincide:

<https://blockchain.info/block-index/106662>  
<https://blockchain.info/block-index/106692>

## 8 Conclusion

We have reviewed how to obtain from a model of a bank first a model of the blockchain based on a simple ledger, and then in a second step a model of the blockchain based on transaction trees. Both models have been modelled in Agda, where we included transaction fees, block rewards, signatures, and transactions based on Merkle trees. We saw that transactions, transaction trees and unspent transaction outputs form an extended form of an inductive-recursive definition.

When studying descriptions of blockchain technology, we found that when it comes to transactions referring to previous unspent transactions, often the material is not presented in a clear way. It seems that the underlying reason is the fact that one needs to deal in some form with the underlying transaction tree. This definition is inductive-recursive in nature, and therefore difficult to understand. We hope that this article will help to obtain a better understanding of the underlying transaction trees.

**What is a bitcoin?** This is a frequently asked question. The correct question is what bitcoins are. If we look at our model we see that the money which can be used are the unspent transaction outputs which can be computed from the current blockchain. So the bitcoins are the unspent transaction outputs of the blockchain, which can be computed from it, where the blockchain is stored in a peer-to-peer network with consensus obtained by the mining protocol.

**Future work.** We have already indicated some aspects still be to be added (mining, input and output scripts in transactions), and indicated what needs to be shown in order to guarantee that the protocol is correct and to prove decidability. Future work would be to use our work with Abel and Adelsberger on writing user interfaces in Agda [1, 2] to write a simulator for Bitcoin in Agda. What would be very challenging is to write a cryptocurrency in Agda, where cryptographic functions make use of the foreign language interface to Haskell available in Agda. One could think of having a language for smart contracts as well, which are verified in Agda. This would allow to have a cryptocurrency in which correctness is proved in the same language as the implementation, and which allows to define verified smart contracts.

---

### References

- 1 Andreas Abel, Stephan Adelsberger, and Anton Setzer. ooAgda. Agda Library, 2016. URL: <https://github.com/agda/ooAgda>.
- 2 Andreas Abel, Stephan Adelsberger, and Anton Setzer. Interactive programming in Agda – Objects and graphical user interfaces. *Journal of Functional Programming*, 27, Jan 2017. URL: <https://doi.org/10.1017/S0956796816000319>, doi:10.1017/S0956796816000319.
- 3 Agda Community. Welcome to Agda’s documentation!, retrieved 27 November 2017. URL: <https://agda.readthedocs.io/>.
- 4 Agda Wiki. The Agda Wiki, 2017. URL: <http://wiki.portal.chalmers.se/agda/pmwiki.php>.
- 5 Marcin Andrychowicz, Stefan Dziembowski, Daniel Malinowski, and Łukasz Mazurek. Modeling Bitcoin contracts by timed automata. In Axel Legay and Marius Bozga, editors, *Formal Modeling and Analysis of Timed Systems: 12th International Conference, FORMATS 2014, Florence, Italy, September 8-10, 2014. Proceedings*, pages 7–22, Cham, 2014. Springer International Publishing. URL: [https://doi.org/10.1007/978-3-319-10512-3\\_2](https://doi.org/10.1007/978-3-319-10512-3_2), doi:10.1007/978-3-319-10512-3\_2.

- 6 Martijn Bastiaan. Preventing the 51%-attack: A stochastic analysis of two phase proof of work in Bitcoin, January 23 2015. 22nd Twente Student conference on IT, January 23, 2015, Enschede, The Netherlands. URL: <http://referaat.cs.utwente.nl/conference/22/paper/7473/preventingthe-51-attack%20-a-stochastic-analysis-of-two-phase-proof-of-work-in-bitcoin.pdf>.
- 7 Mike Belshe. [Bitcoin-segwit2x] Segwit2x final steps, 8 Nov 2017. Email on Bitcoin-segwit2x email list. URL: <https://lists.linuxfoundation.org/pipermail/bitcoin-segwit2x/2017-November/000685.html>.
- 8 W. J. B. Beukema. Formalising the Bitcoin protocol, 2014. URL: <http://referaat.cs.utwente.nl/conference/21/paper/7450/formalising-the-bitcoin-protocol.pdf>.
- 9 Karthikeyan Bhargavan, Antoine Delignat-Lavaud, Cédric Fournet, Anitha Gollamudi, Georges Gonthier, Nadim Kobeissi, Natalia Kulatova, Aseem Rastogi, Thomas Sibut-Pinote, Nikhil Swamy, and Santiago Zanella-Béguelin. Formal verification of smart contracts: Short paper. In *Proceedings of the 2016 ACM Workshop on Programming Languages and Analysis for Security, PLAS '16*, pages 91–96, New York, NY, USA, 2016. ACM. URL: <http://doi.acm.org/10.1145/2993600.2993611>, doi:10.1145/2993600.2993611.
- 10 bitcoin.stackexchange. Can an output be spent in the block in which it is contained?, retrieved 15 Nov 2017. URL: <https://bitcoin.stackexchange.com/questions/30485/can-an-output-be-spent-in-the-block-in-which-it-is-contained>.
- 11 Kaylash Chaudhary, Ansgar Fehnker, Jaco van de Pol, and Mariëlle Stoelinga. Modeling and verification of the Bitcoin protocol. In *Proceedings Workshop on Models for Formal Analysis of Real Systems, MARS 2015, Suva, Fiji, November 23, 2015.*, pages 46–60, 2015. Proceedings are on Arkiv. Number arXiv:1511.04173v1 [cs.LO] <https://arxiv.org/abs/1511.04173>. URL: <http://dx.doi.org/10.4204/EPTCS.196.5>, doi:10.4204/EPTCS.196.5.
- 12 coinmarketcap. Cryptocurrency market capitalization, retrieved 27 Nov 2017. URL: <https://coinmarketcap.com/>.
- 13 Coq Community. The Coq Proof Assistant, 2015. URL: <https://coq.inria.fr/>.
- 14 Peter Dybjer. A general formulation of simultaneous inductive-recursive definitions in type theory. *Journal of Symbolic Logic*, 65(2):525 – 549, June 2000. URL: <https://doi.org/10.2307/2586554>, doi:10.2307/2586554.
- 15 Peter Dybjer and Anton Setzer. Indexed induction-recursion. In Reinhard Kahle, Peter Schroeder-Heister, and Robert Stärk, editors, *Proof Theory in Computer Science*, volume 2183 of *Lecture Notes in Computer Science*, pages 93–113. Springer, 2001. URL: [http://dx.doi.org/10.1007/3-540-45504-3\\_7](http://dx.doi.org/10.1007/3-540-45504-3_7), doi:10.1007/3-540-45504-3\_7.
- 16 Peter Dybjer and Anton Setzer. Induction-recursion and initial algebras. *Annals of Pure and Applied Logic*, 124:1 – 47, 2003. URL: <http://www.sciencedirect.com/science/article/pii/S0168007202000969>, doi:10.1016/S0168-0072(02)00096-9.
- 17 github. bips/bip-0034.mediawiki, 30 Nov 2016. 959fecc. URL: <https://github.com/bitcoin/bips/blob/master/bip-0034.mediawiki>.
- 18 github. Handle historic transactions with duplicate IDs. #42, retrieved 19 Nov 2017. URL: <https://github.com/bitpay/insight-api/issues/42>.
- 19 Damon Jones. Bitcoin: A formal verification and analysis of social aspects and security, May 2014. Third year project, Dept. of Computer Science, Swansea University, Academic year 2013/14.
- 20 A. Kosba, A. Miller, E. Shi, Z. Wen, and C. Papamanthou. Hawk: The blockchain model of cryptography and privacy-preserving smart contracts. In *2016 IEEE Symposium on Security and Privacy (SP)*, pages 839–858, May 2016. doi:10.1109/SP.2016.55.
- 21 Ahmed Kosba, Andrew Miller, Elaine Shi, Zikai Wen, and Charalampos Papamanthou. Hawk: The blockchain model of cryptography and privacy-preserving smart contracts.

- Cryptology ePrint Archive, Report 2015/675, 2015. URL: <https://eprint.iacr.org/2015/675>.
- 22 Loi Luu, Duc-Hiep Chu, Hrishi Olickel, Prateek Saxena, and Aquinas Hobor. Making smart contracts smarter. In *Proceedings of the 2016 ACM SIGSAC Conference on Computer and Communications Security, CCS '16*, pages 254–269, New York, NY, USA, 2016. ACM. URL: <http://doi.acm.org/10.1145/2976749.2978309>, doi:10.1145/2976749.2978309.
  - 23 Per Martin-Löf. *Intuitionistic type theory*. Bibliopolis, Naples, 1984.
  - 24 Ralph C. Merkle. A digital signature based on a conventional encryption function. In Carl Pomerance, editor, *Advances in Cryptology — CRYPTO '87: Proceedings*, pages 369–378, Berlin, Heidelberg, 1988. Springer Berlin Heidelberg. URL: [https://doi.org/10.1007/3-540-48184-2\\_32](https://doi.org/10.1007/3-540-48184-2_32), doi:10.1007/3-540-48184-2\_32.
  - 25 Satoshi Nakamoto. Bitcoin: A peer-to-peer electronic cash system, 1 November 2008. Announced on the cryptography mailing list. URL: <http://www.bitcoin.org/bitcoin.pdf>.
  - 26 Satoshi Nakamoto. Bitcoin P2P e-cash paper, 1 November 2008. Email at the cryptography mailing list. URL: <http://satoshi.nakamotoinstitute.org/emails/cryptography/1/>.
  - 27 Ulf Norell. *Towards a practical programming language based on dependent type theory*. PhD thesis, Department of Computer Science and Engineering, Chalmers University of Technology, SE-412 96 Göteborg, Sweden, September 2007. URL: <http://www.cse.chalmers.se/~ulfn/papers/thesis.pdf>.
  - 28 reddit. What is a TXID? Forum post, retrieved 19 Nov 2017. URL: [https://www.reddit.com/r/Bitcoin/comments/5waqc1/what\\_is\\_a\\_txid/](https://www.reddit.com/r/Bitcoin/comments/5waqc1/what_is_a_txid/).
  - 29 Anton Setzer. bitcoinAgda, 28 November 2017. Agda Library. <https://github.com/csetzer/bitcoinAgda>. URL: <https://github.com/csetzer/bitcoinAgda>.
  - 30 Anton Setzer. Modelling bitcoins in Agda, May 2017. Talk given at 23rd International Conference on Types for Proofs and Programs TYPES 2017, Budapest, Hungary, 29 May - 1 June 2017, Abstract: <http://types2017.elte.hu/proc.pdf>, Slides: <https://www.dropbox.com/sh/unwi307a12pwyuh/AAcTRaRU1H1b1CbldGbqitf9a?dl=0#>.
  - 31 Aaron Stump. *Verified Functional Programming in Agda*. Morgan & Claypool, 2016. URL: <https://svn.divms.uiowa.edu/repos/clc/projects/agda/book/book.pdf>.
  - 32 Nick Szabo. Smart contracts: Building blocks for digital markets, 1996. URL: [http://www.fon.hum.uva.nl/rob/Courses/InformationInSpeech/CDROM/Literature/L0Twinterschool2006/szabo.best.vwh.net/smart\\_contracts\\_2.html](http://www.fon.hum.uva.nl/rob/Courses/InformationInSpeech/CDROM/Literature/L0Twinterschool2006/szabo.best.vwh.net/smart_contracts_2.html).
  - 33 Brian Warner. Bitcoin: A technical introduction, 21 July 2011. Slides and video of brown-bag technical presentation presented at Mozilla. URL: <http://www.lothar.com/presentations/bitcoin-brownbag/>.
  - 34 Wikipedia. Bitcoin cash, retrieved 27 November 2017. URL: [https://en.wikipedia.org/wiki/Bitcoin\\_Cash](https://en.wikipedia.org/wiki/Bitcoin_Cash).
  - 35 Wikipedia. The DAO (organization), retrieved 20 October 2017. URL: [https://en.wikipedia.org/wiki/The\\_DAO\\_\(organization\)](https://en.wikipedia.org/wiki/The_DAO_(organization)).
  - 36 Wikipedia. SegWit, retrieved 2 Dec 2017. Available from <https://en.wikipedia.org/wiki/SegWit>. URL: <https://en.wikipedia.org/wiki/SegWit>.
  - 37 Wikipedia. SetWit2x, retrieved 27 November 2017. URL: <https://en.wikipedia.org/wiki/SegWit2x>.
  - 38 Wikipedia. Smart contract, retrieved 27 November 2017. URL: [https://en.wikipedia.org/wiki/Smart\\_contract](https://en.wikipedia.org/wiki/Smart_contract).